


```
(%i1) kill(all)$;
```

1 Vetknutý nosník částečně zatížený spojitým zatížením

1.1 Zadání

Figure 1:



Zatížení, rozměry, materiál, atd...

Předpokládám nosník kruhového průřezu s průměrem D. Nosník je z oceli.

```
(%i1) cisla:[q=10, l=1.5, E=2.1*10^11, D=0.010, J=%pi * 0.010^4 / 64]$;
```

1.2 Řešení

1.2.1 Rozdělení nosníku na úseky, zavedení souřadnice

Zvedu si souřadnici x od volného (pravého) konce a definuji si oba dva úseky na nosníku.

Pravý úsek označím jako I, levý interval jako II

```
(%i3) usek_I:((x>=0) and (x < l/2))$;  
      usek_II:(x>=l/2 and (x<=l))$;
```

1.2.2 Vnitřní statické účinky - vztahy a grafy

Ohybový moment v prvním a druhém úseku pomocí metody řezu.

Posouvající sílu dostanu pomocí Schwedlerovy věty jako derivaci momentu.

Protože x míří zprava, musím použít Schwedlerovu větu zprava

```
(%i7) M_I: -1/2 * q * x^2$;  
      M_II: -q * l/2 * (x-l/4)$;  
  
      T_I: -diff(M_I,x)$;  
      T_II: -diff(M_II,x)$;
```

Moment má na každém úseku jinou definici (M_I a M_II), já teď obě definice spojujím do jediné definice veličiny M a totéž udělám s posouvající silou T


```
(%i9) M: (if usek_I then M_I elseif usek_II then M_II);
      T: (if usek_I then T_I elseif usek_II then T_II);
```

$$(M) \quad \text{if } x \geq 0 \text{ and } x < \frac{l}{2} \text{ then } -\frac{q x^2}{2} \text{ elseif } x \geq \frac{l}{2} \text{ and } x \leq l \text{ then } -\frac{l q \left(x - \frac{l}{4}\right)}{2}$$

$$(T) \quad \text{if } x \geq 0 \text{ and } x < \frac{l}{2} \text{ then } q x \text{ elseif } x \geq \frac{l}{2} \text{ and } x \leq l \text{ then } \frac{l q}{2}$$

Vykreslení průběhu momentu a posouvající síly.

```
(%i10) wxplot2d(ev([M,T], cisla), [x,0,ev(l, cisla)], [legend, "M_o","T"] );
(%t10)
```




```
(%o10)
```

Protože ve skutečnosti máme x zprava a Maxima nám (podle očekávání) vykreslí graf s x zleva, musíme si s tím nějak poradit. Můžeme to udělat například tak, že si zavedeme souřadnici x_1 , měřenou zleva. Každý bod nosníku má tedy souřadnici x měřenou zprava a souřadnici x_1 měřenou zleva a platí $x+x_1=l$. Takže $x=l-x_1$. Postačí tedy za x dosadit $(l-x_1)$ a dostaneme vztahy pro moment a posouvající sílu vyjádřené pomocí x_1 měřeného zleva. Tyto vztahy poslouží pro vykreslení grafu:

```
(%i11) wxplot2d(ev([M,T], cisla, x=ev(l, cisla)-x1), [x1,0,ev(l, cisla)], [legend, "M_o","T"] );
```

(%t11)



(%o11)

1.2.3 Průhybová čára

Zapíšu diferenciální rovnici průhybové čáry pro první i pro druhý úsek

```
(%i13) dr1:'diff(w_I,x,2)=-M_I/(E * J);  
dr2:'diff(w_II,x,2)=-M_II/(E * J);
```

$$(dr1) \quad \frac{d^2}{dx^2} w_I = \frac{q x^2}{2 E J}$$

$$(dr2) \quad \frac{d^2}{dx^2} w_{II} = \frac{l q \left(x - \frac{l}{4} \right)}{2 E J}$$

Příkazem ode2 každou rovnici vyřeším

```
(%i15) reseni_I:ode2(dr1, w_I, x);  
reseni_II:ode2(dr2, w_II, x);
```

$$(reseni_I) \quad w_I = \frac{q x^4}{24 E J} + \%k2 x + \%k1$$

$$(reseni_II) \quad w_{II} = \frac{4 l q x^3 - 3 l^2 q x^2}{48 E J} + \%k2 x + \%k1$$

Ve skutečnosti integrační konstanta %k1 z prvního řešení není totéž jako %k1 z druhého řešení. Podobně %k2 z obou řešení také nejsou shodné.

Jako průhybovou čáru v prvním úseku vezmu tedy pravou stranu (příkaz rhs) řešení v prvním úseku a zároveň konstanty %k1 a %k2 přejmenuji na C1, C2. Totéž udělám se druhým úsekem, kde konstanty přejmenuji na C3 a C4. Vypočítám také sklony jako derivace průhybových čar. Protože x je zprava, musím před derivaci doplnit znaménko mínus.

```
(%i19) w_I:ev(rhs(reseni_I), %k1=C1, %k2=C2);
      w_II:ev(rhs(reseni_II),%k1=C3, %k2=C4);
```

```
fi_I:-diff(w_I, x);
fi_II:-diff(w_II,x);
```

$$(w_I) \quad \frac{q x^4}{24 E J} + C2 x + C1$$

$$(w_II) \quad \frac{4 l q x^3 - 3 l^2 q x^2}{48 E J} + C4 x + C3$$

$$(fi_I) \quad -\frac{q x^3}{6 E J} - C2$$

$$(fi_II) \quad -\frac{12 l q x^2 - 6 l^2 q x}{48 E J} - C4$$

Zapíšu si všechny okrajové a přechodové podmínky:

- shodný průhyb prvního a druhého úseku na souřadnici $x=l/2$
- shodný sklon prvního a druhého úseku na souřadnici $x=l/2$
- nulový průhyb úseku II na souřadnici $x=l$ (ve vetknutí)
- nulový sklon úseku II na souřadnici $x=l$ (ve vetknutí)

```
(%i23) op1:ev(w_I,x=l/2) = ev(w_II,x=l/2);
      op2:ev(fi_I,x=l/2) = ev(fi_II,x=l/2);
      op3:ev(w_II,x=l) = 0;
      op4:ev(fi_II,x=l) = 0;
```

$$(op1) \quad \frac{l^4 q}{384 E J} + \frac{C2 l}{2} + C1 = -\frac{l^4 q}{192 E J} + \frac{C4 l}{2} + C3$$

$$(op2) \quad -\frac{l^3 q}{48 E J} - C2 = -C4$$

$$(op3) \quad \frac{l^4 q}{48 E J} + C4 l + C3 = 0$$

$$(op4) \quad -\frac{l^3 q}{8 E J} - C4 = 0$$

Okrajové podmínky tvoří soustavu rovnic, z níž vyřešíme hodnoty konstant....

```
(%i24) konstanty:solve([op1,op2,op3,op4],[C1,C2,C3,C4])[1];
```

$$(konstanty) \quad [C1 = \frac{41 l^4 q}{384 E J}, C2 = -\frac{7 l^3 q}{48 E J}, C3 = \frac{5 l^4 q}{48 E J}, C4 = -\frac{l^3 q}{8 E J}]$$

...a ty dosadíme do vztahů pro průhyby a sklony

```
(%i28) w_I:ev(w_I, konstanty);
      w_II:ev(w_II, konstanty);
```

```
      fi_I:ev(fi_I, konstanty);
      fi_II:ev(fi_II, konstanty);
```

$$(w_I) \quad \frac{q x^4}{24 E J} - \frac{7 l^3 q x}{48 E J} + \frac{41 l^4 q}{384 E J}$$

$$(w_II) \quad \frac{4 l q x^3 - 3 l^2 q x^2}{48 E J} - \frac{l^3 q x}{8 E J} + \frac{5 l^4 q}{48 E J}$$

$$(fi_I) \quad \frac{7 l^3 q}{48 E J} - \frac{q x^3}{6 E J}$$

$$(fi_II) \quad \frac{l^3 q}{8 E J} - \frac{12 l q x^2 - 6 l^2 q x}{48 E J}$$

Protože máme průhyb a sklon definován ve dvou úsecích, musíme je sloučit podobně, jako jsme to udělali s momentem a posouvající silou

```
(%i30) w:(if usek_I then w_I elseif usek_II then w_II);
      fi:(if usek_I then fi_I elseif usek_II then fi_II);
```

$$(w) \quad \text{if } x \geq 0 \text{ and } x < \frac{l}{2} \text{ then } \frac{q x^4}{24 E J} - \frac{7 l^3 q x}{48 E J} + \frac{41 l^4 q}{384 E J} \text{ elseif } x \geq \frac{l}{2} \text{ and } x \leq l \text{ then } \frac{4 l q x^3 - 3 l^2 q x^2}{48 E J} - \frac{l^3 q x}{8 E J} + \frac{5 l^4 q}{48 E J}$$

$$(fi) \quad \text{if } x \geq 0 \text{ and } x < \frac{l}{2} \text{ then } \frac{7 l^3 q}{48 E J} - \frac{q x^3}{6 E J} \text{ elseif } x \geq \frac{l}{2} \text{ and } x \leq l \text{ then } \frac{l^3 q}{8 E J} - \frac{12 l q x^2 - 6 l^2 q x}{48 E J}$$


Pro účely vykreslení grafu dosadím do vztahů čísla

```
(%i31) grafy:ev([w, fi], ciska)$;
```

...a nakonec vykreslím grafy s pomocí už jednou použitého fíglu se souřadnicí x1 zleva

```
(%i32) wxplot2d(ev(grafy, x=ev(l, ciska)-x1), [x1,0, ev(l, ciska)], [legend, "w", "fi"] );
```

(%t32)



(%o32)

Spočítáme ještě průhyb a sklon na konci. Průhyb bude v milimetrech, sklon ve stupních

```
(%i33) float(ev([w*1000,fi*180/%pi],ciska,x=0));  
(%o33) [52.43586964366909,2.73567195834312]
```

Created with [wxMaxima](#).